# AE 342 : Modeling and Analysis Lab, Session 4

Gaurav Gupta, SC21B026, Aerospace Engineering

November 20, 2023

## 1 Problem 1

A simple 1D heat diffusion problem with Dirichlet boundary conditions is simulated. The governing equations and boundary conditions are provided as follows

$$\frac{d^2u}{dx^2} = 1, 0 < x < 1 \tag{1}$$

$$u(x = 0) = 0,\ u(x = 1) = 0 \tag{2}$$

The problem is solved in FreeFEM++ using the given code for a 1D domain.

```
load "msh3"
real m = 100;
int l=1;
meshL Th = segment(m,[x*l]);
real[int] xaxis(m+1), Uline(m+1);
ofstream file2("Results.csv");

fespace Vh(Th, P1);
Vh U, v;

solve Poission(U,v)= int1d(Th)(dx(U)*dx(v))
        +int1d(Th)(v)
        +on(1,2, U=0);

plot(U, value=true);
for(int i=0; i<=m; i++){

  xaxis[i] = i/m;
  Uline[i] = U[][i];
  file2 << xaxis[i] << "," << Uline[i] << endl ;
}
```

Listing 1: Problem 1 Code

The analytical solution for the problem is given as $U = \frac{1}{2}(x^2 - x)$. The solution of FreeFEM is compared to the analytical solution,




Figure 1: Comparision of the FreeFEM Solution with analytical solution for Problem 1.

## 2 Problem 2

Consider a 1mm diameter, 50mm long aluminium pin-fin. The fin' s left end interfaces with a constant heat source at a temperature of $T_h = 300°C$, while its right end is thermally insulated. Heat is progressively released along the fin's length through convection, interacting with an ambient temperature of $T_\infty = 30°C$. The governing differential equation is given by:

$$k\frac{d^2T}{dx^2} = \frac{Ph}{A}(T - T_\infty) \tag{3}$$

In equation 3, the thermal conductivity (k) is 200 W/mK and the convective heat transfer coefficient (h) is $20W/m^2K$. P, A represents perimeter and area of cross-section of the fin. Establish a model for the temperature distribution along the fin under these specific conditions and compare with analytical results.

The weak form of the equation 3 is given as

$$\int_\phi \left( k\frac{dT}{dx}\frac{dv}{dx} + \frac{Ph}{A}(T - T_\infty) \right) dx = 0 \tag{4}$$

The problem is solved in FreeFEM++ using the given code for a 1D domain.

```
load "msh3"
real m = 100;
real d = 0.001;
int h = 20;
int k = 200;
real l=0.05;
int Ti = 300;
int Tf = 30;
real p = pi * d;
real A = pi * d * d / 4;

meshL Th = segment(m,[x*l]);

real[int] xaxis(m+1), Uline(m+1);
ofstream file2("results2.csv");

fespace Vh(Th, P1);
Vh T, v;

solve Poission(T,v)  = int1d(Th)(k*dx(T)*dx(v))
        +int1d(Th)(p*h/A*T*v)
        -int1d(Th)(p*h/A*Tf*v)
        +on(1, T=300);

plot(T, value=true, wait=true, fill=true, aspectratio=true);
for(int i=0; i<=m; i++){

  xaxis[i] = l * i/m;
  Uline[i] = T[][i];
  file2 << xaxis[i] << "," << Uline[i] << endl ;
}
```

Listing 2: Problem 2 Code

The analytical solution for the problem is given as $T = T_\infty + \left(\frac{270}{1+e^{2mL}}\right)\left(e^{mx} + e^{2ml-mx}\right)$. The solution of FreeFEM is compared to the analytical solution in Figure 2.

When the right end is subjected to the ambient temperature $T_\infty = 30$ instead of the adiabatic condition. The analytical solution of the problem is given as $T = T_\infty + \left(\frac{-270}{1-e^{2mL}}\right)\left(e^{mx} - e^{2ml-mx}\right)$. The result of this problem solved in FreeFEM is again compared to the analytical solution in Figure 3.

```
solve Poission(T,v)  = int1d(Th)(k*dx(T)*dx(v))
        +int1d(Th)(p*h/A*T*v)
        -int1d(Th)(p*h/A*Tf*v)
        +on(1,T=300)
        +on(2,T=30);
```

Listing 3: Change in the Problem 2 Code to accomodate for the Dirchlet B.C. at the right end




Figure 2: Comparision of the FreeFEM Solution with analytical solution for Problem 2.




Figure 3: Comparision of the FreeFEM Solution with analytical solution for Problem 2 with Dirchlet boundary condition at right end.

## 3 Problem 3

Consider a rectangular plate of dimensions $L_x \times L_y$, where one side is insulated, while the other three sides are maintained at constant temperatures. The temperature distribution $T(x, y)$ in the plate is governed by Laplace's equation:

$$\nabla^2 T = 0 \tag{5}$$

**Boundary Conditions:**

- At $x = 0$: $\frac{\partial T}{\partial x} = 0$
- At $x = L_x : T(L_x, y) = 200°C$
- At $y = 0 : T(x, 0) = 150°C$
- At $y = L_y : T(x, L_y) = 150°C$

The weak formulation of the laplace equation is given as,

$$\int_\phi \left( \frac{dT}{dx}\frac{dv}{dx} + \frac{dT}{dy}\frac{dv}{dy} \right) dx = 0 \tag{6}$$

```
load "msh3"

real Lx = 1;
real Ly = 2;
int m = 100;
int n = 200;
mesh Th=square(m,n,[Lx*x,Ly*y]);
//plot(Th, wait=true);
//real[int] V = [150, 155, 160, 165, 170, 175, 180, 185, 190, 195, 200];

fespace Vh(Th, P1);
Vh T, v;

solve Laplace(T,v) = int2d(Th)(dx(T)*dx(v) + dy(T)*dy(v))
        +on(1,3,T=150)
        +on(2,T=200);

plot(T, wait=true, fill=true, value=true, aspectratio=true);
```

Listing 4: Code to solve Problem 3 in FreeFEM

The temperature distirbution $T(x, y)$ for the problem 3 is plotted in Figure 4.




Figure 4: temperature distirbution $T(x, y)$ for the problem 3 plotted in FreeFEM.